



Base de données

Nous utiliserons une des bases de données exemples fournies avec Visual Basic et vue en cours (base de données bibliographique).

Emplacement de la base de données sur le disque dur (il n'est pas nécessaire de dupliquer la base) :

C:\Program Files\DevStudio\VB\Biblio.mdb

Schéma relationnel de la base :

Publishers (PubID, Name, CompanyName, Address, City, State, Zip, Telephone, Fax, Comments)

Titles (ISBN, Title, YearPublished, PubID, Description, Notes, Subject, Comments)

Authors (Au_ID, Author, YearBorn)

TitleAuthor (ISBN, Au_ID)

Clés primaires Clés étrangères

Requête disponible : AllTitles (liste des ouvrages avec le nom de la compagnie qui les publie)

Application VB

L'objectif du TD est de réaliser une application qui permette de faire des recherches de références bibliographiques par mot clé (par exemple, rechercher tous les ouvrages traitant de Visual Basic).

Contrôles obligatoires

- Un contrôle Donnée (*Data*) nommé *Titres* connecté à la requête *AllTitles* de la base de données *Biblio.mdb*.
- Une Zone de texte (*TextBox*) nommée *Mot* pour la saisie du ou des mots clés.
- Une Zone de liste classique (*ListBox*) nommée *Liste* pour l'affichage du résultat de la requête (au format « Titre, N° ISBN, Auteur, Année, Éditeur »).
- Un Bouton de commande (*CommandButton*) nommé *Rechercher* pour déclencher la recherche. Pour écrire le code associé à ce bouton, il est possible de s'inspirer de l'étude de cas vue en cours (ftp://seeco-nte.univ-lyon2.fr/Darmont/maitrise/cas_vbbd.pdf – Recherche par auteur).

Méthodes du contrôle Donnée *Titres* à utiliser : *Recordset.FindFirst()*
Recordset.FindNext()

Propriétés du contrôle Donnée *Titres* à utiliser : *Recordset.NoMatch*
Recordset.Fields().value

Utiliser l'aide en ligne pour obtenir des détails sur le fonctionnement de ces méthodes et propriétés.


Contrôles optionnels

- Un Bouton de commande (*CommandButton*) nommé *Nouvelle* dont le code permettra de désactiver (propriété *Enabled*) la liste *Liste* et le bouton *Nouvelle* lui-même, ainsi que d'activer la zone de texte *Mot* et le bouton *Rechercher*. De façon « symétrique », le code du bouton *Rechercher* devra être modifié pour, en plus de la recherche, désactiver la zone de texte *Mot* et le bouton *Rechercher* et activer la liste *Liste* et le bouton *Nouvelle*.
- Un Bouton de commande (*CommandButton*) nommé *Quitter* dont le code permettra, ô surprise, de quitter l'application.

Question subsidiaire : Amélioration de l'application

Telle qu'elle, la recherche par mot clé renvoie plusieurs lignes pour un ouvrage lorsque ce dernier a été écrit par plusieurs auteurs. Modifier le code du bouton *Rechercher* de manière à n'afficher qu'une ligne par ouvrage, mais en indiquant tous les auteurs, séparés par des virgules. Il est possible de s'inspirer de l'étude de cas vue en cours (Catalogue complet).

Annexe : Apparence de l'application



Correction

```
Private Sub Rechercher_Click()  
  
    ' Déclaration des données à récupérer  
    Dim title As String  
    Dim isbn As String  
    Dim isbn2 As String  
    Dim author As String  
    Dim year As String  
    Dim pub As String  
    ' Ligne du résultat  
    Dim ligne As String  
    ' Condition SQL (recherche)  
    Dim condition As String  
  
    ' Désactivation de la recherche  
    Mot.Enabled = False  
    Rechercher.Enabled = False  
    ' Activation du résultat  
    liste.Enabled = True  
    Nouvelle.Enabled = True  
    ' Nettoyage de la liste  
    liste.Clear  
  
    ' Recherche dans la table  
    condition = "Title like '*' + Mot.Text + '*'"  
    Titres.Recordset.FindFirst (condition) ' Recherche 1er  
    While Not Titres.Recordset.NoMatch ' Tant que trouvé  
        ' Lecture des données  
        title = Titres.Recordset.Fields("Title").Value  
        isbn = Titres.Recordset.Fields("ISBN").Value  
        author = Titres.Recordset.Fields("Author").Value  
        year = CStr(Titres.Recordset.Fields("Year Published").Value)  
        pub = Titres.Recordset.Fields("Company Name").Value  
        ligne = title + ", " + isbn + ", " + author + ", " + year + ", " + pub  
        liste.AddItem (ligne) ' Ajout dans la liste  
        Titres.Recordset.FindNext (condition) ' Au suivant !  
    Wend  
  
End Sub  
  
Private Sub Nouvelle_Click()  
    ' Désactivation du résultat  
    liste.Enabled = False  
    Nouvelle.Enabled = False  
    ' Activation de la recherche  
    Mot.Enabled = True  
    Rechercher.Enabled = True  
    Mot.Text = ""  
  
End Sub  
  
Private Sub Quitter_Click()  
    End  
End Sub
```

```
Private Sub Rechercher_Click() ' Avec élimination des doublons  
  
    ' Déclaration des données à récupérer  
    Dim title As String  
    Dim isbn As String  
    Dim isbn2 As String  
    Dim author As String  
    Dim year As String  
    Dim pub As String  
    ' Ligne du résultat  
    Dim ligne As String  
    ' Condition SQL (recherche)  
    Dim condition As String  
  
    ' Désactivation de la recherche  
    Mot.Enabled = False  
    Rechercher.Enabled = False  
    ' Activation du résultat  
    liste.Enabled = True  
    Nouvelle.Enabled = True  
    ' Nettoyage de la liste  
    liste.Clear  
  
    ' Recherche dans la table  
    condition = "Title like '*' + Mot.Text + '*'"  
    Titres.Recordset.FindFirst (condition) ' Recherche 1er  
    While Not Titres.Recordset.NoMatch ' Tant que trouvé  
        ' Lecture des données  
        title = Titres.Recordset.Fields("Title").Value  
        isbn = Titres.Recordset.Fields("ISBN").Value  
        year = CStr(Titres.Recordset.Fields("Year Published").Value)  
        pub = Titres.Recordset.Fields("Company Name").Value  
        ligne = title + ", " + isbn  
        ' Tant que l'article est le même, on ajoute les auteurs  
        isbn2 = isbn  
        While (isbn2 = isbn) And (Not Titres.Recordset.EOF)  
            author = Titres.Recordset.Fields("Author").Value  
            ligne = ligne + ", " + author  
            Titres.Recordset.MoveNext ' Enregistrement suivant  
            If Not Titres.Recordset.EOF Then  
                isbn2 = Titres.Recordset.Fields("ISBN").Value  
            End if  
        Wend  
        ' On complète la ligne  
        ligne = ligne + ", " + year + ", " + pub  
        liste.AddItem (ligne) ' Ajout dans la liste  
        Titres.Recordset.FindNext (condition) ' Au suivant !  
    Wend  
  
End Sub
```